Phys 238: Histogram of Pendulum Periods

In[1]:=
```
Clear["Global`*"]
```

In[2]:=
```
SetDirectory[NotebookDirectory[]];
```

In[13]:=
```
FilePrint["pendulum-20250129.csv", 2] (* There are 3 columns; we want the third. *)
```

```
"Latest: Time (s)","Latest: GateState","Latest: Period (s)"

0.139623,1,
```

In[14]:=
```
fulldata = Import["pendulum-20250129.csv", "CSV"];
```

In[15]:=
```
data = Select[fulldata[[All, 3]], NumberQ];
```

In[16]:=
```
Histogram[data]
```

Out[16]=



In[17]:=
```
μ = Mean[data]
σ = StandardDeviation[data]
```

Out[17]=

```
1.93466
```

Out[18]=

```
0.000168487
```

Now try to model this distribution by a Gaussian. Getting a manual list of the histogram numbers is a bit tedious. The HistogramList[ ] function actually gives two separate lists, first is a list of bin boundaries, and second is the list of counts in each bin.

In[19]:=
```
{bins, counts} = N[HistogramList[data]];
```

Interleaving them is straightforward, if verbose. We will assign each height to the center point of each bin by constructing a new table.

In[20]:=
```
hist = Table[{(bins[[i]] + bins[[i + 1]]) / 2., counts[[i]]},
   {i, 1, Length[counts]}];
```

Enter an unnormalized gaussian (Eq. 5.20 in Taylor's text):

In[21]:=
```
gauss[A_, μ_, σ_, x_] := A/(σ Sqrt[2 π]) Exp[-(x - μ)^2/(2 σ^2)]
```

$$\texttt{gauss[A\_, }\mu\texttt{\_, }\sigma\texttt{\_, x\_] := }\frac{A}{\sigma\,\texttt{Sqrt}[2\,\pi]}\,\texttt{Exp}\left[-\frac{(x-\mu)^2}{2\,\sigma^2}\right]$$

In[22]:=
```
Clear[A, μ, σ]
(* The initial fit attempt does always work well.  It will be useful to
 seed the process with guesses at the mean and standard deviation. *)
result = NonlinearModelFit[hist, gauss[A, μ, σ, x],
  {A,
   {μ, Mean[data]},
   {σ, StandardDeviation[data]}
  }, x]
```

Out[23]=

FittedModel[$61.1\, e^{-1.75\times10^{7} \ll 1 \gg}$]

In[24]:=
```
result["BestFitParameters"]
```

Out[24]=

{A → 0.0258702, μ → 1.93467, σ → 0.000169052}

In[25]:=
```
result["ParameterConfidenceIntervalTable"]
```

Out[25]=

	Estimate	Standard Error	Confidence Interval
A	0.0258702	0.000726064	{0.0243603, 0.0273801 }
μ	1.93467	5.47782×10^{-6}	{1.93466, 1.93468 }
σ	0.000169052	5.48004×10^{-6}	{0.000157656 , 0.000180449 }

These are very close to the calculated mean and standard deviation:

In[26]:=
```
{Mean[data], StandardDeviation[data]}
```

Out[26]=

{1.93466, 0.000168487}

Show the histogram, the data used in the fit, and the best-fit gaussian all on the same graph. Add useful titles.

In[27]:=
```
histoplot = Show[{
    Histogram[data],
    ListPlot[hist, PlotStyle → Black],
    Plot[result[x], {x, Min[bins], Max[bins]}]},
  Frame → True, FrameLabel → {"Period", "Count"},
  LabelStyle → Larger, ImageSize → Scaled[0.8]]
```

Out[27]=



In[28]:=
```
Export["pendulum-histo-plot.pdf", histoplot]
```

Out[28]=

pendulum-histo-plot.pdf

Physical Pendulum vs Simple Pendulum



$$I = ML^2 + \frac{1}{12} M h^2 \qquad \text{(parallel axis)}$$

$$= ML^2 \left[1 + \frac{1}{12} \left(\frac{h}{L} \right)^2 \right]$$

$$T = 2\pi \sqrt{\frac{I}{mgL}}$$

For a point mass, $I = mL^2$

$$T = 2\pi \sqrt{\frac{L}{g_{pt}}} \Rightarrow g_{pt} = \frac{4\pi^2}{T^2} L$$

For a cylinder,

$$T = 2\pi \sqrt{\frac{I}{m g_{cyl} L}} \Rightarrow g_{cyl} = \frac{4\pi^2}{T^2} \cdot \frac{I}{mL}$$

$$g_{cyl} = \frac{4\pi^2}{T^2} \cdot \frac{mL^2\left(1 + \frac{1}{12}\frac{h^2}{L^2}\right)}{mL}$$

$$g_{cyl} = \frac{4\pi^2}{T^2} \cdot L \left(1 + \frac{1}{12}\frac{h^2}{L^2}\right)$$

$$g_{cyl} - g_{pt} = \underbrace{\frac{4\pi^2}{T^2} \cdot L}_{\text{this is } g_{pt}} \left(\frac{1}{12}\frac{h^2}{L^2}\right)$$

$$g_{cyl} - g_{pt} = g_{pt}\left[\frac{1}{12}\frac{h^2}{L^2}\right]$$

Estimates $h = 8\ \text{cm}$

$L = 100\ \text{cm}$

$$\frac{1}{12}\frac{h^2}{L^2} = 5.3 \times 10^{-4}$$

Assuming $g_{pt} = 9.8\ m/s^2$,

$$g_{cyl} - g_{pt} = 0.005\ m/s^2$$

Compare that to your overall uncertainty, typically $\sim 1\ \text{mm}$ in L

$$\frac{\delta L}{L} \sim \frac{1}{1000} = 10^{-3},$$

$$\delta g \simeq g_{pt}\left(\frac{\delta L}{L}\right) = 0.0098 \simeq 0.01$$

∴ Correction due to physical pendulum is about half of the typical uncertainty due to the uncertainty in Length.

Further corrections ?

these are small corrections on top of the small correction due to the physical pendulum.